Problem:
Score : 700 points

Problem Statement

We have a sandglass consisting of two bulbs, bulb A and bulb B. These bulbs contain some amount of sand.
When we put the sandglass, either bulb A or B lies on top of the other and becomes the upper bulb. The other bulb becomes the lower bulb.

The sand drops from the upper bulb to the lower bulb at a rate of 1 gram per second.
When the upper bulb no longer contains any sand, nothing happens.

Initially at time 0, bulb A is the upper bulb and contains a grams of sand; bulb B contains X-a grams of sand (for a total of X grams).

We will turn over the sandglass at time r_1,r_2,..,r_K. Assume that this is an instantaneous action and takes no time. Here, time t refer to the time t seconds after time 0.

You are given Q queries.
Each query is in the form of (t_i,a_i).
For each query, assume that a=a_i and find the amount of sand that would be contained in bulb A at time t_i.

Constraints

1≤X≤10^9

1≤K≤10^5

1≤r_1<r_2< .. <r_K≤10^9

1≤Q≤10^5

0≤t_1<t_2< .. <t_Q≤10^9

0≤a_i≤X (1≤i≤Q)

All input values are integers.

Input

The input is given from Standard Input in the following format:

X
K
r_1 r_2 .. r_K
Q
t_1 a_1
t_2 a_2
:
t_Q a_Q

Output

For each query, print the answer in its own line.

Sample Input 1

180
3
60 120 180
3
30 90
61 1
180 180

Sample Output 1

60
1
120

In the first query, 30 out of the initial 90 grams of sand will drop from bulb A, resulting in 60 grams.
In the second query, the initial 1 gram of sand will drop from bulb A, and nothing will happen for the next 59 seconds. Then, we will turn over the sandglass, and 1 second after this, bulb A contains 1 gram of sand at the time in question.

Sample Input 2

100
1
100000
4
0 100
90 100
100 100
101 100

Sample Output 2

100
10
0
0

In every query, the upper bulb initially contains 100 grams, and the question in time comes before we turn over the sandglass.

Sample Input 3

100
5
48 141 231 314 425
7
0 19
50 98
143 30
231 55
342 0
365 100
600 10

Sample Output 3

19
52
91
10
58
42
100

Solution in Fortran (gfortran v4.8.4) (judge verdict: Accepted)

program sandglass
  implicit none
  integer :: x, k, q, a(100000), tt(200000,2), ans(100000), i
  integer :: tlft, trit, alft, arit, t0, t1, s
  logical :: rev
  a = 0
  tt = 0
  ans = 0
  read(*,*) x
  read(*,*) k
  read(*,*) tt(1:k,1)
  read(*,*) q
  do i = 1, q
    read(*,*) tt(k+i,1), a(i)
    tt(k+i,2) = i
  end do
  call merge_sort(tt(1:k+q,1:2))
  tlft = 0
  trit = x
  alft = 0
  arit = x
  rev = .false.
  t0 = 0
  do i = 1, k+q
    t1 = tt(i,1)
    s = tt(i,2)
    alft = alft-(t1-t0)
    if (alft.lt.0) then
      tlft = tlft-alft
      if (tlft.gt.x) tlft = x
      alft = 0
    end if
    arit = arit-(t1-t0)
    if (arit.lt.0) then
      trit = trit+arit
      if (trit.lt.0) trit = 0
      arit = 0
    end if
    if (tlft.ge.trit) then
      if (rev) then
        tlft = trit-1
        if (tlft.lt.0) tlft = 0
      else
        trit = tlft+1
        if (trit.gt.x) trit = x
      end if
    end if
    !write(*,'(6(i0,x))') t0, t1, tlft, trit, alft, arit
    t0 = t1
    if (s.eq.0) then
      rev = .not.rev
      alft = x-alft
      arit = x-arit
      cycle
    end if
    if (a(s).le.tlft) then
      ans(s) = alft
    else if (a(s).ge.trit) then
      ans(s) = arit
    else
      ans(s) = a(s)-tlft+alft
      if (rev) ans(s) = -a(s)+tlft+alft
    end if
    if (rev) ans(s) = x-ans(s)
  end do
  do i = 1, q
    write(*,'(i0)') ans(i)
  end do
  stop
contains
  subroutine merge_sort(c)
    implicit none
    integer, intent(inout) :: c(:,:)
    integer :: n, m, l, u, i
    n = size(c(:,1))
    m = n
    l = 1
    do while (m.gt.1)
      do i = 1, m/2
        u = min(2*i*l,n)
        call merger(c(2*(i-1)*l+1:(2*i-1)*l,:),c((2*i-1)*l+1:u,:))
      end do
      l = 2*l
      m = (m+1)/2
    end do
    return
  end subroutine merge_sort
  subroutine merger(c1,c2)
    implicit none
    integer, intent(inout) :: c1(:,:), c2(:,:)
    integer :: c(size(c1(:,1))+size(c2(:,1)),size(c2(1,:)))
    integer :: i1, i2, n1, n2
    i1 = 1
    i2 = 1
    n1 = size(c1(:,1))
    n2 = size(c2(:,1))
    do while ((i1.le.n1).and.(i2.le.n2))
      if (c1(i1,1).le.c2(i2,1)) then
        c(i1+i2-1,:) = c1(i1,:)
        i1 = i1+1
      else
        c(i1+i2-1,:) = c2(i2,:)
        i2 = i2+1
      end if
    end do
    if (i1.le.n1) then
      c(i1+i2-1:n1+n2,:) = c1(i1:n1,:)
    else if (i2.le.n2) then
      c(i1+i2-1:n1+n2,:) = c2(i2:n2,:)
    end if
    c1 = c(1:n1,:)
    c2 = c(n1+1:n1+n2,:)
    return
  end subroutine merger
end program sandglass